New Features in Mplus Version 8.9 and Forthcoming 8.10: Part 1

Bengt Muthén

Professor Emeritus, UCLA
Mplus: https://www.statmodel.com
bmuthen@statmodel.com

I thank Tihomir Asparouhov for helpful advice and Thuy Nguyen and Noah Hastings for expert assistance.

ASEM AESEM BSEM

DSEM RDSEM ESEM

Mplus®

RSEM PSEM PSEM-ELGM

ALF/LASSO/GEOMIN

Workshop Schedule

8:30-10:00	Lecture 1
10:00-10:15	Q & A
— 15 minute break —	
10:30-12:00	Lecture 2
12:00-12:15	Q & A
— 1 hour lunch break —	
1:15-2:45	Lecture 3
2:45- 3:00	Q & A
— 15 minute break —	
3:15-4:45	Lecture 3
4:45-5:00	Q & A

Outline

- Modeling with residual variables (hat variables):
  - Growth modeling with residual auto-regressions
  - RI-CLPM
  - Predicting from residuals
- Longitudinal factor analysis
  - Longitudinal ESEM factor analysis
  - Longitudinal measurement invariance testing
  - Longitudinal alignment
- Cross-sectional factor analysis
  - BSEM and PSEM

Example: Positive Affect Across the Week Days

- Average PA per day for 7 days, N = 244
  - Data from an EMA study with approximatly 8 measurements per day at random times (used in Mplus Web Talk 6 on DSEM)
  - Thanks are due to Loes Keijsers, PI, who provided the data (the data set cannot be shared)
  - Dietvost et al. (2021). Grumpy of depressed? Disentangling typically developing adolescent mood from prodromal depression using experience sampling methods. Journal of Adolescence.
- PA is the average of 6 items:
  - Low arousal: Relaxed, satisfied, confident
  - High arousal: Happy, energetic, excited
- Covariates:
  - Time-invariant: Gender, SDQ (childhood emotional problems)
  - Time-varying: Tiredness

Modeling with Residual Variables

- Asparouhov & Muthén (2023). Residual Structural Equation Models. Structural Equation Modeling: A Multidisciplinary Journal - see Recent Papers on the Mplus home page
- Yˆ (Y-hat) representing the residual in regressions such as Y ON X or F BY Y
- Hats available only for single-level models (except with RDSEM)
- ML and WLSMV estimation using the Theta parameterization: Convenience feature
- Bayes estimation: Special algorithm (Bayes does not like variances fixed at zero)
- 3 application areas: Growth modeling, RI-CLPM, predicting from residuals

Growth Modeling where Adjacent Residuals have Correlations (WITH) or Auto-Regressions (ON)





- Auto-regression is preferrable and needs hat language to refer to the residuals (WITH refers to residuals of DVs; see WT4P1, slide 13)
  - UG ex 6.17 uses MODEL CONSTRAINT for auto-regressions which is cumbersome except for small T and univariate model
- Growth mixture modeling can also use hats for auto-regression

Growth Modeling of PA: 7 Days, Tuesday - Monday



- Cubic growth?

Mplus Input for Auto-Regressive Residuals Using Hats




```
           USEVARIABLES = pa1-pa7;
ANALYSIS:
           ESTIMATOR = MLR;
MODEL:     i s q c | pa1@0 pa2@.1 pa3@.2 pa4@.3
           pa5@.4 pa6@.5 pa7@.6;
           q@0; c@0;
           pa2ˆ-pa7ˆ PON pa1ˆ-pa6ˆ (ar);
OUTPUT:
           STANDARDIZED TECH4;
PLOT:
           TYPE = PLOT3;
           SERIES = pa1-pa7(s);
```

TECH4 Correlations for the Residuals

	PA1^	PA2^	PA3^	PA4^	PA5^	PA6^	PA7^
PA1^	1.000						
PA2^	0.316	1.000					
PA3^	0.114	0.360	1.000				
PA4^	0.040	0.128	0.356	1.000			
PA5^	0.011	0.034	0.095	0.268	1.000		
PA6^	0.004	0.011	0.031	0.088	0.330	1.000	
PA7^	0.001	0.005	0.013	0.037	0.138	0.417	1.000

- Residual correlations are not zero for lags 2, 3, ... as they are for the adjacent correlation approach

Model Testing Results (N = 244)

Model	# par's	logL	BIC	χ^2	Df	P-value	RMSEA	Prob < 0.05
No AR	14	-1335	2747	48	21	0.0006	0.073	0.077
AR	15	-1319	2720	30	20	0.0766	0.044	0.576
Rescorr	15	-1322	2726	32	20	0.0408	0.050	0.462

- AR fit advantage over Rescorr would be larger for a larger sample size

Estimated Growth Factor Means for Model with AR

Parameter	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
I	5.757	0.056	102.536	0.000
S	-1.192	0.560	-2.128	0.033
Q	7.377	2.486	2.968	0.003
C	-9.152	2.791	-3.279	0.001



Outline

- Modeling with residual variables (hat variables):
  - Growth modeling with residual auto-regressions
  - **RI-CLPM**
  - Predicting from residuals
- Longitudinal factor analysis
  - Longitudinal ESEM factor analysis
  - Longitudinal measurement invariance testing
  - Longitudinal alignment
- Cross-sectional factor analysis
  - BSEM and PSEM

RI-CLPM

- Hamaker, Kuiper, Grasman (2015). A critique of the cross-lagged panel model. Psychological Methods
- Hamaker (2023). The within-between dispute in cross-lagged panel research and how to move forward. Forthcoming in Psych Methods



(a) Old approach using factors



(b) New approach using hats

Mplus Input for RI-CLPM: Old versus New

```
           USEVARIABLES = y1-y5 z1-z5;
ANALYSIS:  ESTIMATOR = ML;
           MODEL = NOCOV;
MODEL:     iy BY y1-y5@1;
           iz BY z1-z5@1;
           iy WITH iz;
           ! Defining within factors:
           fy1 BY y1;
           fy2 BY y2;
           fy3 BY y3;
           fy4 BY y4;
           fy5 BY y5;
           fz1 BY z1;
           fz2 BY z2;
           fz3 BY z3;
           fz4 BY z4;
           fz5 BY z5;
           y1-z5@0;
           ! AR:
           fy2-fy5 PON fy1-fy4;
           fz2-fz5 PON fz1-fz4;
           ! Cross-lags
           fy2-fy5 PON fz1-fz4;
           fz2-fz5 PON fy1-fy4;
           Residual covariances:
           fy1-fy5 PWITH fz1-fz5;
```

```
           USEVARIABLES = y1-y5 z1-z5;
ANALYSIS:  ESTIMATOR = ML;
MODEL:     iy BY y1-y5@1;
           iz BY z1-z5@1;
           ! AR:
           y2^-y5^ PON y1^-y4^;
           z2^-z5^ PON z1^-z4^
           ! Cross-lags
           y2^-y5^ PON z1^-z4^;
           z2^-z5^ PON y1^-y4^;
           Residual covariances:
           y1^-y5^ PWITH z1^-z5^;
```


RI-CLPM with Factors: Model Alternative 1

- Mulder & Hamaker (2020). Three extensions of the Random Intercept Cross-Lagged Panel Model. Structural Equation Modeling



RI-CLPM with Factors: Model Alternative 2



- Hat language can be applied to the factors: f2^-f5^ PON f1^-f4^

RI-CLPM with a Categorical Outcome and Multiple Groups

- Mplus Web Talk 4, Part 2
- Estimators: WLSMV, Bayes (ML gets too many dimensions of integration)
- Models: Probit, Two-part ordinal
- Mplus defaults:
  - Residual variances for categorical outcomes fixed at 1
- Multiple-group analysis:
  - Which parameters should be tested for equality across groups?
    - Cross-lagged effects
- Bayes can do Wald testing using MODEL TEST
  - Asparouhov & Muthén (2021). Advances in Bayesian model fit evaluation for structural equation models, Structural Equation Modeling

Mplus Input for 9-Group RI-CLPM with a Categorical Outcome Using Bayes

- 8 treatment groups and a placebo group are represented as 9 classes using KNOWNCLASS in a TYPE = MIXTURE analysis using Bayesian estimation (WT4P2, slide 154 -, discussing tx effects)
- Y continuous, Z binary

```
           USEVARIABLES = y1-y8 z1-z8;
           CATEGORICAL = z1-z8;

           CLASSES = c(9);
           KNOWNCLASS = c(cCell = 1-9);

ANALYSIS:  TYPE = MIXTURE;
           ESTIMATOR = BAYES;
           BITERATIONS = (5000);
           THIN = 10;
           PROCESSORS = 8;
```

```
MODEL:   %OVERALL%
         iy BY y1-y8@1;
         iz BY z1-z8@1;

         y2^-y8^ PON y1^-y7^;
         z2^-z8^ PON z1^-z7^;

         y2^-y8^ PON z1^-z7^;
         z2^-z8^ PON y1^-y7^;

         y1^-y8^ PWITH z1^-z8^;

OUTPUT:  STANDARDIZED  RESIDUAL
         TECH1 TECH8 TECH10;
```

- Intercepts and thresholds held equal across classes (groups) as the default. RI factor means free except fixed at zero for last class (group)

9-Group RI-CLPM: Testing Group Differences

```
MODEL:        %OVERALL%
              iy BY y1-y8@1;
              iz BY z1-z8@1;

              y2^-y8^ PON y1^-y7^;
              z2^-z8^ PON z1^-z7^;

              ! Time-invariant cross-lags:
              y2^-y8^ PON z1^-z7^ (yz);
              z2^-z8^ PON y1^-y7^ (zy);

              y1^-y8^ PWITH z1^-z8^;

              %c#1% ! 1st group is placebo
              y2^-y8^ PON z1^-z7^ (yzp);
              z2^-z8^ PON y1^-y7^ (zyp);

MODEL TEST:   ! Placebo vs all other groups:
              ! 2 df Wald test
              0 = yz - yzp;
              0 = zy - zyp;
```


Predicting from Residuals




```
MODEL:
         f BY x1-x4;
         y ON f x4^;
```

Outline

- Modeling with residual variables (hat variables):
  - Growth modeling with residual auto-regressions
  - RI-CLPM
  - Predicting from residuals
- **Longitudinal factor analysis**
  - Longitudinal ESEM factor analysis
  - Longitudinal measurement invariance testing
  - Longitudinal alignment
- Cross-sectional factor analysis
  - BSEM and PSEM

ESEM Theory in Brief

- Asparouhov & Muthén (2009). Exploratory structural equation modeling. Structural Equation Modeling: A Multidisciplinary Journal
- Application papers at: `http://www.statmodel.com/ESEM.shtml`
- EFA is covered in Mplus Short Courses, Topic 1 (video and handout)

EFA Factor Indeterminacy And Rotations

- $\Sigma = \Lambda \Psi \Lambda^T + \Theta$
- Λ *is p x m, so* $\Lambda \Psi \Lambda^T$ *has* m^2 indeterminacies.
  - $m = 1 : \lambda^2 \psi = \lambda^{*2} \psi^*$ *for* $\lambda^* = \lambda / \sqrt{c}, \psi^* = \psi c$
- $\Psi = I$ fixes m(m+1)/2 indeterminacies (m(m-1)/2 remaining)
- This gives $\Lambda \Lambda^T + \Theta = \Lambda^* \Lambda^{*^T} + \Theta$ where Λ^* is the rotated Λ, $\Lambda^* = \Lambda H^{-1}$, where H is orthogonal, i.e., $H^T = H^{-1}$, which means that $H^{-1} H^{-1^T} = I$.
- Example: Unrotated model for m = 2:

$$\Psi = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \Lambda = \begin{bmatrix} X & 0 \\ X & X \\ X & X \\ X & X \\ X & X \\ X & X \end{bmatrix} \quad (1)$$

- A starting Λ can be rotated using a rotation criterion function such as Geomin that favors a simple Λ structure while allowing correlated factors - the simple structure criterion adds the needed information to avoid the indeterminacies

Factor Loading Rotation in EFA: P = 6 (Dots), M = 2



Unrotated solution
$\lambda_{12} = 0$
$\Psi = I$

Orthogonal rotation
90 degrees: Uncorr factors

Oblique rotation
Correlated factors

Transformation Of SEM Parameters Based On Rotated Λ

Measurement part:
(1) $Y_i = \nu + \Lambda\eta_i + KX_i + \varepsilon_i$

Structural part:
(2) $\eta_i = \alpha + B\,\eta_i + \Gamma X_i + \xi_i$

Transformations:

(6) $\nu^* = \nu$

(7) $\Lambda^* = \Lambda H^{-1}$

(8) $K^* = K$

(9) $\Theta^* = \Theta$

(10) $\alpha^* = H\alpha$

(11) $B^* = HBH^{-1}$

(12) $\Gamma^* = H\Gamma$

(13) $\Psi^* = H'\Psi H$

Example: MIMIC ESEM



- Step 1: Unrotated loadings model, estimating Λ ($\Psi = I$), and the Γ regressions of factors on covariates
- Step 2: EFA rotation of y measurement part gives new Λ, Ψ and the transformation is applied to the estimated Γ from step 1

Example: Longitudinal Exploratory Factor Analysis of Positive Affect Across 7 Week Days

- PA average per day for 7 days (Tue - Mon): N = 244, T = 7
  - Dietvost et al. (2021). Grumpy of depressed? Disentangling typically developing adolescent mood from prodromal depression using experience sampling methods. Journal of Adolescence.
- PA is the average of 6 items:
  - Low arousal: Relaxed, satisfied, confident
  - High arousal: Happy, energetic, excited
- What's the factor structure for the 6 items?
  - EFA is needed
- Separate analyses of each day has disadvantages:
  - Using less information
  - Confounding trait with state variation
  - Cattell-Molenaar-Hamaker-Steyer-Eid-Geiser (LST theory)

Longitudinal Factor Analysis Model 3 Indicators of 1 Factor at 2 Time Points



- 3 building blocks:
  - Random intercepts (i1-i3)
  - Auto-regressions for Within factors (f2 ON f1)
  - Auto-regressions among indicator-specific residuals may be needed (not drawn)
- PA example: 6 indicators of 2 factors at 7 time points

PA Example: 6 Indicators of 2 Factors (2 Timepoints Shown)



- One process in the RI-CLPM factor figure or both using same items?
- With 6 variables, you can have have max 3 factors in EFA:
  - Within-factor EFA calls for ESEM to handle the full model
- Covariates:
  - Time-invariant: Gender, SDQ influencing random intercepts
  - Time-varying: Tiredness influencing within factors

PA Example: 6 Indicators, 2 FW, 1 FB



- Between factor model also for the 6 random intercepts (CFA or EFA)
- Between and within factors have possibly different factor structures

Longitudinal ESEM Factor Analysis: Input

```
            USEVARIABLES = relax1-excit7;
            ! 6 PA items, 3 low arousal 3 high arousal:
            ! relaxed (pala1) satisfied (pala2) confident (pala3)
            ! happy (paha1) energetic (paha2) excited (paha3)
            ! 7 time points

ANALYSIS:   ESTIMATOR = MLR;

MODEL:      ! random intercepts for the 6 items
            i1 BY relax1-relax7@1;
            i2 BY satis1-satis7@1;
            i3 BY conf1-conf7@1;
            i4 BY happy1-happy7@1;
            i5 BY energ1-energ7@1;
            i6 BY excit1-excit7@1;

            ! auto-regressions among factor indicators residuals:
            relax2^-relax7^ PON relax1^-relax6^ (ar1);
            satis2^-satis7^ PON satis1^-satis6^ (ar2);
            conf2^conf7^ PON conf1^-conf6^ (ar3);
            happy2^-happy7^ PON happy1^-happy6^ (ar4);
            energ2^-energ7^ PON energ1^-energ6^ (ar5);
            excit2^-excit7^ PON excit1^-excit6^ (ar6);
```


Longitudinal ESEM Factor Analysis Input Cont'd

```
! 2-factor ESEM with metric (loading) invariance as in UG ex 5.26
! (factor variances at first time point are automatically fixed at 1):
f11-f12 BY relax1 satis1 conf1 happy1 energ1 excit1(*1 1);
f21-f22 BY relax2 satis2 conf2 happy2 energ2 excit2(*2 1);
f31-f32 BY relax3 satis3 conf3 happy3 energ3 excit3(*3 1);
f41-f42 BY relax4 satis4 conf4 happy4 energ4 excit4(*4 1);
f51-f52 BY relax5 satis5 conf5 happy5 energ5 excit5(*5 1);
f61-f62 BY relax6 satis6 conf6 happy6 energ6 excit6(*6 1);
f71-f72 BY relax7 satis7 conf7 happy7 energ7 excit7(*7 1);

! auto-regressions among factors to reduce the number of parameters
f21-f22 ON f11-f12; f31-f32 ON f21-f22; f41-f42 ON f31-f32;
f51-f52 ON f41-f42; f61-f62 ON f51-f52; f71-f72 ON f61-f62;
i1-i6 WITH f11-f72@0;

! scalar invariance for intercepts:
[relax1-relax7] (int1);
[satis1-satis7] (int2);
[conf1-conf7] (int3);
[happy1-happy7] (int4);
[energ1-energ7] (int5);
[excit1-excit7] (int6);
[f11-f12@0 f21-f72*];
```

ESEM with Metric/Scalar Invariance as Part of a Bigger Model: How Does it Work?

- The unrotated model is estimated holding the unrotated loadings equal across time, fixes factor variances at 1 for the first time point, identifies factor covariance at the first time point, and unrestricted factor covariance matrices at all other time points (2 factors involve 2 par's decided by rotation: TECH1 shows 2 more par's than in the results)
- The unrotated factors can have an AR model
- The rotation transformation is then applied to the entire model
- The necessary restriction is that if a variable is regressed on a factor in an EFA block it has to be regressed on all factors in the EFA block - the same applies for the opposite regression
- CFA factors (like random intercepts) can be combined with EFA factors
  - If one EFA factor is correlated with the CFA factor all other EFA factors in the same block must also be correlated (random intercepts are uncorrelated with the ESEM factors)

Standardized Factor Loading Estimates for Metric Model

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
F11 BY				
RELAX1	**0.471**	0.052	9.072	0.000
SATIS1	**0.550**	0.061	8.999	0.000
CONF1	**0.352**	0.055	6.379	0.000
HAPPY1	**0.274**	0.046	5.984	0.000
ENERG1	0.016	0.030	0.540	0.589
EXCIT1	-0.007	0.002	-3.331	0.001
F12 BY				
RELAX1	0.001	0.017	0.078	0.938
SATIS1	-0.007	0.023	-0.312	0.755
CONF1	0.023	0.037	0.623	0.533
HAPPY1	**0.308**	0.051	6.089	0.000
ENERG1	**0.475**	0.053	8.901	0.000
EXCIT1	**0.613**	0.055	11.107	0.000
F11 WITH F12				
	0.616	0.094	6.573	0.000

- Unlike EFA, ESEM analyzes a sample covariance matrix - although the factors at t=1 have variance 1, the outcomes don't: Stand'd values correspond to EFA

Further Results

- Factor indicator-specific residual AR1:
  - Significant for 3 out of the 6 items: Confident, happy, energetic
- Lag1 regressions among the two within factors:
  - AR1 significant for only high-arousal PA factor
  - Insignificant cross-lagged effects between the two factors
- Percentage variance explained by the random intercepts
- Variance decompositions, reliability:
  - Eid et al. (2017). On the definition of latent-state-trait models with autoregressive effects. European Journal of Psychological Assessment, 33, 285-295

Adding Other Variables

- Validating the need for two factors: Is there an important difference between how the two factors relate to other variables?
  - Do they have different relations to covariates or distal outcomes?
  - Such other variables can be included in longitudinal ESEM
- The time-varying covariate tired is included in the current example

Longitudinal ESEM Factor Analysis with a Time-Varying Covariate: Additional Input in the Overall MODEL

```
f11-f12 ON tired1;
f21-f22 ON f11-f12 tired2;
f31-f32 ON f21-f22 tired3;
f41-f42 ON f31-f32 tired4;
f51-f52 ON f41-f42 tired5;
f61-f62 ON f51-f52 tired6;
f71-f72 ON f61-f62 tired7;
tired1-tired7; ! to avoid deletion due to missing data
```


Longitudinal ESEM Factor Analysis with the Time-Varying Covariate Tired

Factor Model			Regressing the factors on time varying tiredness covariate		
	PALow	PAHigh		PALow	PAHigh
RELAXED	**0.49**	0.02	t = 1 (Tue)	-0.23	**-0.62**
SATISFIED	**0.60**	-0.03	t = 2 (Wed)	-0.03	**-0.26**
CONFIDENT	**0.37**	0.03	t = 3 (Thur)	-0.13	**-0.43**
HAPPY	**0.28**	**0.38**	t = 4 (Fri)	**-0.51**	**-0.46**
ENERGETIC	-0.03	**0.65**	t = 5 (Sat)	-0.17	**-0.45**
EXCITED	0.02	**0.67**	t = 6 (Sun)	-0.21	**-0.56**
Correlation	0.67		t = 7 (Mon)	-0.11	**-0.37**

Between-Level Factor Model (Metric Within Factor Model)

Model	# par's	logL	BIC	χ^2	Df	P-value	RMSEA	Prob < 0.05
1-factor	155	-6527	13907	1321	790	0.0000	0.053	0.200
2-factor (EFA in CFA)	160	-6511	13901	1299	785	0.00000	0.052	0.271
Unrestricted	164	-6500	13901	1281	781	0.0000	0.051	0.338

- 2*logL difference (df) indicates rejection of 1- and 2-factor models:
  - 1-factor vs unrestricted: 54 (df = 9)
  - 2-factor vs unrestricted: 22 (df = 4), 2-factor correlation = 0.84
  - 1-factor vs 2-factor: 32 (df = 5)
  - 2-factor CFA equal loadings between and within: 26 (df = 5), $p < 0.01$
- 2-factor EFA in CFA (no ESEM on factors yet in Mplus): Fix at zero the i6 loading for the first factor and the i1 loading for the second factor while fixing factor variances at 1 ($m^2 = 4$ Lambda-Psi restrictions)

Standardized PA Factor Loadings on Between and Within with EFA-in-CFA for Between and ESEM for Within

Between EFA in CFA		Within EFA (ESEM)	
Low Arousal	High Arousal	Low Arousal	High Arousal
0.94	0*	**0.47**	-0.00
0.94	0.07	**0.56**	-0.01
0.61	0.25	**0.36**	0.02
0.50	**0.51**	**0.28**	**0.31**
-0.03	**0.99**	0.02	**0.48**
0*	**1.00**	-0.01	**0.61**

- * Fixed value
- Bolded values are significant
- Factor correlations are 0.84 versus 0.63
- The 2 between factors have similar regression coefficients for gender (females lower) and SDQ (childhood emotional problems; lower)

Longitudinal Factor Analysis Chi-Square Testing: Sample Size, Number of Variables, Number of Time Points

- P factor indicators per time point and T time points result in:
  - P*T variables = 6*7 = 42 in our example
  - The H0 model may have more parameters than the sample size
  - The H1 model has P*T*(P*T+1)/2 = 903 parameters in our example
- What is the quality of the regular chi-square testing of the H0 model against the unrestricted H1 model?
- Simulations based on the estimated model suggest inflated chi-square 5% reject proportions:
  - N = 250: 0.41
  - N = 500: 0.14
  - N = 1000: 0.11
  - Parameter estimates, SEs, and coverage good even at smaller N
- Small sample sizes may not be able to handle large P*T:
  N > P*T is needed as a bare minimum

Outline

- Modeling with residual variables (hat variables):
  - Growth modeling with residual auto-regressions
  - RI-CLPM
  - Predicting from residuals
- Longitudinal factor analysis
  - Longitudinal ESEM factor analysis
  - **Longitudinal measurement invariance testing**
  - Longitudinal alignment
- Cross-sectional factor analysis
  - BSEM and PSEM

Longitudinal Measurement Invariance Testing

- Type of models:
  - CFA
  - EFA/ESEM as part of a larger model such as the longitudinal factor analysis model with covariates just discussed
- Estimators:
  - ML and WLSMV
  - Not Bayes (no ESEM, no chi-square difference testing)
- Model specification:
  - ANALYSIS command using the MODEL option:
    MODEL = CONFIGURAL METRIC SCALAR
  - Overall MODEL describing relationships across time
  - T time-specific MODEL t defining factors for each time point for which measurement invariance is tested

Longitudinal Measurement Invariance Testing

```
          USEVARIABLES = relax1-excit7;
          ! 6 PA items, 3 low arousal, 3 high arousal:
          ! relaxed (pala1) satisfied (pala2) confident (pala3)
          ! happy (paha1) energetic (paha2) excited (paha3)
          ! 7 time points

ANALYSIS: ESTIMATOR = MLR;
          MODEL = CONFIGURAL METRIC SCALAR;

MODEL:    ! random intercepts for all 6 items:
          i1 BY relax1-relax7@1;
          i2 BY satis1-satis7@1;
          i3 BY conf1-conf7@1;
          i4 BY happy1-happy7@1;
          i5 BY energ1-energ7@1;
          i6 BY excit1-excit7@1;
          ! auto-regressions among factor indicators residuals:
          relax2^-relax7^ PON relax1^-relax6^ (ar1);
          satis2^-satis7^ PON satis1^-satis6^ (ar2);
          conf2^conf7^ PON conf1^-conf6^ (ar3);
          happy2^-happy7^ PON happy1^-happy6^ (ar4);
          energ2^-energ7^ PON energ1^-energ6^ (ar5);
          excit2^-excit7^ PON excit1^-excit6^ (ar6);
```

Longitudinal Measurement Invariance Testing Cont'd

```
            ! AR1 regressions among factors
            ! to reduce the number of parameters:
            f21-f22 ON f11-f12;
            f31-f32 ON f21-f22;
            f41-f42 ON f31-f32;
            f51-f52 ON f41-f42;
            f61-f62 ON f51-f52;
            f71-f72 ON f61-f62;

            i1-i6 with f11-f72@0;

            ! 2-factor ESEM for each time point:
MODEL t1:   f11-f12 by relax1 satis1 conf1 happy1 energ1 excit1(*1);
MODEL t2:   f21-f22 by relax2 satis2 conf2 happy2 energ2 excit2(*2);
MODEL t3:   f31-f32 by relax3 satis3 conf3 happy3 energ3 excit3(*3);
MODEL t4:   f41-f42 by relax4 satis4 conf4 happy4 energ4 excit4(*4);
MODEL t5:   f51-f52 by relax5 satis5 conf5 happy5 energ5 excit5(*5);
MODEL t6:   f61-f62 by relax6 satis6 conf6 happy6 energ6 excit6(*6);
MODEL t7:   f71-f72 by relax7 satis7 conf7 happy7 energ7 excit7(*7);
```


Longitudinal Measurement Invariance Test Results

Model	Number of Parameters	χ^2	Degrees of Freedom	P-Value
Configural	212	1246.811	733	0.0000
Metric	164	1281.246	781	0.0000
Scalar	140	1326.299	805	0.0000

Models Compared	χ^2	Degrees of Freedom	P-Value
Metric against Configural	57.341	48	0.1673
Scalar against Configural	94.980	72	0.0362
Scalar against Metric	46.660	24	0.0037

- The scalar model is typically rejected but is needed for growth modeling
- Ways out of this dilemma include:
  - Longitudinal alignment
  - Approximate invariance using BSEM or PSEM, e.g. for only the scalar part (intercepts)

Automated Approach Advantages

- Also handles categorical outcomes with Delta and Theta parameterizations
- Automatically uses the scaling correction factors for chi-square difference testing with MLR and uses DIFFTEST with WLSMV

Longitudinal Measurement Invariance Testing with Categorical Outcomes: WLSMV

- Three 8-category ordinal items measuring 1 factor at 7 time points

```
           USEVARIABLES = bkThin1f bkThin1s bkThin2s
           bkThin3s bkThin4s bkThin5s bkThin6s
           harmO1f harmO1s harmO2s harmO3s harmO4s harmO5s
           harmO6s
           takeP1f takeP1s takeP2s takeP3s takeP4s TakeP5s takeP6s ;

           CATEGORICAL = bkThin1f - takeP6s;
           MISSING = ALL (999);

ANALYSIS:  ESTIMATOR = WLSMV;
           MODEL = CONFIGURAL SCALAR;
```

Input Continued

```
MODEL:       ! As before, but optional

MODEL t1:    f1 BY bkthin1f
             harmo1f
             takeP1f ;
MODEL t2:    f2 BY bkthin1s
             harmo1s
             takeP1s ;
MODEL t3:    f3 BY bkthin2s
             harmo2s
             takeP2s ;
MODEL t4:    f4 BY bkthin3s
             harmo3s
             takeP3s ;
MODEL t5:    f5 BY bkthin4s
             harmo4s
             takeP4s ;
MODEL t6:    f6 BY bkthin5s
             harmo5s
             takeP5s;
MODEL t7:    f7 BY bkthin6s
             harmo6s
             takeP6s ;
```


Different Number of Categories for Different Time Points

- Adding one more time point leads to an error:
  - *** ERROR in MODEL command MODEL T1 and MODEL T8 are not equivalent. The categorical indicators in the same position for factors across time must have the same number of categories. Problem with: BKTHIN1F and BKTHIN7S
- With ML and Bayes, this can be handled by the * approach:
  - CATEGORICAL = bkThin1f - takeP7s(*);
- WLSMV cannot handle the * approach
- ML can handle it but requires numerical integration and there are typically too many dimensions of integration due to many factors
- Bayes can handle it and is feasible, but no measurement invariance chi-square testing summary is provided

Outline

- Modeling with residual variables (hat variables):
  - Growth modeling with residual auto-regressions
  - RI-CLPM
  - Predicting from residuals
- Longitudinal factor analysis
  - Longitudinal ESEM factor analysis
  - Longitudinal measurement invariance testing
  - **Longitudinal alignment**
- Cross-sectional factor analysis
  - BSEM and PSEM

Alignment Papers

- Multiple-group alignment:
  - Asparouhov & Muthén (2014). Multiple-group factor analysis alignment. Structural Equation Modeling: A Multidisciplinary Journal, 21:4, 495-508.
  - Muthén & Asparouhov (2014). IRT studies of many groups: The alignment method. Frontiers in Psychology
  - Muthén & Asparouhov (2018). Recent methods for the study of measurement invariance with many groups: Alignment and random effects. Sociological Methods & Research, 47:4 637-664.
- Generalized multiple-group alignment - ASEM, AESEM (allowing cross-loadings, ESEM, factors regressed on factors, covariates):
  - Asparouhov & Muthén (2023). Multiple group alignment for exploratory and structural equation models. Structural Equation Modeling: A Multidisciplinary Journal, 30(2), 169-191.
- Longitudinal alignment:
  - Section 5.3 of Asparouhov & Muthén (2023). Penalized structural equation models. Forthcoming in SEM

Alignment Theory Briefly Stated: Multiple Groups

- The alignment model has the same fit as the configural model
- Alignment minimizes the amount of measurement noninvariance in intercepts and loadings by estimating group-varying factor means and variances
- The group-varying factor means and variances are not identified without scalar invariance - alignment avoids this problem by adding the necessary extra information via optimization of a simplicity criterion similar to EFA rotation criteria avoiding indeterminacies

Alignment Theory Briefly Stated: 2 Steps

- 1. Estimate the configural model:
  - Loadings and intercepts free across groups, factor means fixed at zero in all groups, factor variances fixed at 1 in all groups
- 2. Do the alignment optimization:
  - Free the factor means and variances and choose their values to minimize the amount of noinvariance using a simplicity function
- In step 2, the factor means α_j and variances ψ_j are free parameters to be estimated, maintaining the configural model fit while obtaining the aligned λ_j and ν_j for group j:

$$\lambda_j = \lambda_{j,configural}/\sqrt{\psi_j} \quad (2)$$

$$\nu_j = \nu_{j,configural} - \alpha_j\, \lambda_{j,configural}/\sqrt{\psi_j} \quad (3)$$

Why “Alignment”? Intercept Invariance but Factor Diffs

- Intercepts of 10 indicators (dots)
- 1 factor
- 2 groups (axes)
- Factor means = 0, -1
- Factor variances = 1, 2



Unaligned: Configural model (factor mean=0, factor variance=1 in both groups)



Aligned: Taking into account the group differences in factor means and variances

Alignment Application (Muthén & Asparouhov, 2018)



Factor means for 26 countries: Scalar versus Alignment

- Alignment agrees with Scalar that Sweden/Cyprus have the highest/lowest level of tradition nonconformity
- Alignment disagrees with Scalar regarding the difference between Portugal and Netherlands, between France and Switzerland, as well as between Norway and Russia

Alignment Analysis Steps

- 3 alignment settings: Fixed, Free, Fixed (group/timepoint)
- Free preferred unless little noninvariance or SEs are large
- Group/timepoint with smallest factor mean can be chosen for Fixed()

Longitudinal Alignment for PA Factors

```
            USEVARIABLES = relax1-excit7;
            ! 6 PA items, 3 low arousal, 3 high arousal:
            ! relaxed (pala1) satisfied (pala2) confident (pala3)
            ! happy (paha1) energetic (paha2) excited (paha3)

ANALYSIS:   ESTIMATOR = MLR;
            ALIGNMENT = FIXED; ! this is the only change
            ! to the measurement invariance input that used
            ! MODEL = CONFIGURAL etc

MODEL:      ! random intercepts for all 6 items:
            i1 BY relax1-relax7@1;
            i2 BY satis1-satis7@1;
            i3 BY conf1-conf7@1;
            i4 BY happy1-happy7@1;
            i5 BY energ1-energ7@1;
            i6 BY excit1-excit7@1;
            ! auto-regressions among factor indicators residuals:
            relax2^-relax7^ PON relax1^-relax6^ (ar1);
            satis2^-satis7^ PON satis1^-satis6^ (ar2);
            conf2^conf7^ PON conf1^-conf6^ (ar3);
            happy2^-happy7^ PON happy1^-happy6^ (ar4);
            energ2^-energ7^ PON energ1^-energ6^ (ar5);
            excit2^-excit7^ PON excit1^-excit6^ (ar6);
```


Longitudinal Alignment Cont'd

```
            ! AR1 regressions among factors
            ! to reduce the number of parameters:
            f21-f22 ON f11-f12;
            f31-f32 ON f21-f22;
            f41-f42 ON f31-f32;
            f51-f52 ON f41-f42;
            f61-f62 ON f51-f52;
            f71-f72 ON f61-f62;

            i1-i6 with f11-f72@0;

            ! 2-factor ESEM for each time point:
MODEL t1:   f11-f12 by relax1 satis1 conf1 happy1 energ1 excit1(*1);
MODEL t2:   f21-f22 by relax2 satis2 conf2 happy2 energ2 excit2(*2);
MODEL t3:   f31-f32 by relax3 satis3 conf3 happy3 energ3 excit3(*3);
MODEL t4:   f41-f42 by relax4 satis4 conf4 happy4 energ4 excit4(*4);
MODEL t5:   f51-f52 by relax5 satis5 conf5 happy5 energ5 excit5(*5);
MODEL t6:   f61-f62 by relax6 satis6 conf6 happy6 energ6 excit6(*6);
MODEL t7:   f71-f72 by relax7 satis7 conf7 happy7 energ7 excit7(*7);
```


Longitudinal Alignment Results: Measurement Part

APPROXIMATE MEASUREMENT INVARIANCE
(NONINVARIANCE) FOR TIMES

Intercepts/Thresholds							
RELAX1	1	2	3	4	5	6	7
SATIS1	1	2	3	4	5	6	7
CONF1	1	2	3	4	5	6	7
HAPPY1	1	2	3	4	5	6	7
ENERG1	1	2	3	4	5	6	7
EXCIT1	1	2	3	4	5	6	7
Loadings for F11							
RELAX1	1	2	3	4	5	6	7
SATIS1	1	2	3	4	5	6	7
CONF1	1	2	3	4	5	6	7
HAPPY1	1	2	3	4	5	6	7
ENERG1	1	2	3	4	5	6	7
EXCIT1	1	2	3	4	5	6	7
Loadings for F12							
RELAX1	1	2	3	4	5	6	7
SATIS1	1	2	3	4	5	6	7
CONF1	1	2	3	4	5	6	7
HAPPY1	1	2	3	4	5	6	7
ENERG1	1	2	3	4	5	6	7
EXCIT1	1	2	3	4	5	6	7

- No parentheses means no measurement noninvariance

Longitudinal Alignment Results: Factor Means

FACTOR INTERCEPT COMPARISON AT THE
5% SIGNIFICANCE LEVEL IN DESCENDING ORDER

Results for Factor F11

Factor Ranking	Time	Factor Intercept	Times With Significantly Smaller Factor Intercept
1	5	0.299	1 3 7
2	4	0.002	
3	1	0.000	7
4	2	-0.114	
5	6	-0.137	
6	3	-0.152	
7	7	-0.336	

Results for Factor F12

Factor Ranking	Time	Factor Intercept	Times With Significantly Smaller Factor Intercept
1	5	0.300	1 3 6
2	4	0.236	1 3 6
3	7	0.095	6
4	1	0.000	6
5	2	-0.046	
6	3	-0.055	
7	6	-0.302	

Estimated Means for Tue-Mon: Low Arousal PA Factor



Scalar Model



Alignment

- According to the alignment result summary, Sat (5) is significantly higher than Tue (1), Thur (3), Mon (7).
  Tue is significantly higher than Mon
- Scalar and alignment often similar

Estimated Means for Tue-Mon: High Arousal PA Factor



Scalar Model



Alignment

- Alignment result summary: Saturday significantly higher than Tue, Thur, Sun. Fri significantly higher than Tue, Thur, Sun
- This is unlike the observed variable curve which is like the first factor

Outline

- Modeling with residual variables (hat variables):
  - Growth modeling with residual auto-regressions
  - RI-CLPM
  - Predicting from residuals
- Longitudinal factor analysis
  - Longitudinal ESEM factor analysis
  - Longitudinal measurement invariance testing
  - Longitudinal alignment
- **Cross-sectional factor analysis**
  - BSEM and PSEM

Bayesian SEM (BSEM) vs Penalized SEM (PSEM)

- BSEM uses Bayesian estimation with small-variance priors
  - Muthén & Asparouhov (2012). Bayesian SEM: A more flexible representation of substantive theory. Psychological Methods, 17, 313-335
- PSEM uses maximum-likelihood estimation with penalty priors (alignment loss function, ALF)
  - Asparouhov & Muthén (2023). Penalized structural equation models. Forthcoming in SEM.
- Both are used with non-identified models
- Simple examples:
  - Confirmatory factor analysis including all cross-loadings
  - MIMIC model including all direct effects from covariates to factor indicators
- The needed extra information to identify the model comes from informative priors/penalties

BSEM-PSEM Application to PA Factor Analysis with Cross-Loadings using Day 1 Data (N = 215)

	Longitudinal ESEM		Day 1 EFA		Day 1 CFA	
RELAX	**0.47**	0.00	**0.99**	**-0.19**	**0.82**	0
SATIS	**0.55**	-0.01	**0.95**	0.02	**0.98**	0
CONF	**0.35**	0.02	**0.65**	0.13	**0.75**	0
HAPPY	**0.27**	**0.31**	**0.47**	**0.53**	0	**0.94**
ENERG	0.02	**0.48**	0.07	**0.85**	0	**0.89**
EXCIT	-0.01	**0.61**	-0.01	**0.99**	0	**0.94**
Correlation	0.62		0.76		0.83	

- Note the smaller loadings and factor correlation for longitudinal ESEM compared to Day 1 EFA (day 1 EFA chi-square (4) = 1.6 (p = 0.806))
- Day 1 CFA assumes that theory specifies that the first and last 3 items measure different factors, low and high arousal PA: Chi-quare (8) = 50.2 (p = 0.000)
- Note the higher factor correlation for Day 1 CFA compared to Day 1 EFA

BSEM Small-Variance Priors for Cross-Loadings



Day 1 EFA for PA items (N = 215)

	PALow	PAHigh
RELAX	**0.99**	**-0.19**
SATIS	**0.95**	0.02
CONF	**0.65**	0.13
HAPPY	**0.47**	**0.53**
ENERG	0.07	**0.85**
EXCIT	-0.01	**0.99**
Correlation	0.76	

- Strength of hypothesis from low to high: EFA - BSEM - CFA
- EFA: Least specific - number of factors
- CFA: Most specific - number of factors and location of zero loadings
- BSEM: In between - number of factors and key items hypothesized to measure those factors (similar to target rotation)
  - Assume that theory specifies that the first and last 3 items measure different factors, low and high arousal PA

Input for BSEM Factor Analysis with Cross-Loadings

```
                 USEVARIABLES = relax1 satis1 conf1 happy1 energ1 excit1;
                 ! 6 PA items, 3 low affect, 3 high affect:
                 ! relaxed satisfied confident
                 ! happy energetic excited

ANALYSIS:        ESTIMATOR = BAYES;
                 BITERATIONS = (5000);
                 THIN = 10;
MODEL:           ! Key loadings based on theory:
                 flow BY relax1* satis1 conf1;
                 fhigh BY happy1* energ1 excit1;
                 flow-fhigh@1;

                 ! Cross-loadings:
                 flow BY happy1-excit1*0 (a1-a3);
                 fhigh BY relax1-conf1*0 (b1-b3);

MODEL PRIORS:    a1-b3~N(0,0.01);
```

Input for PSEM Factor Analysis with Cross-Loadings

```
                USEVARIABLES = relax1 satis1 conf1 happy1 energ1 excit1;
                ! 6 PA items, 3 low affect, 3 high affect:
                ! relaxed satisfied confident
                ! happy energetic excited

ANALYSIS:       ESTIMATOR = MLR; ! Only MLR SEs

MODEL:          ! Key loadings:
                flow BY relax1* satis1 conf1;
                fhigh BY happy1* energ1 excit1;
                flow-fhigh@1;

                ! Cross-loadings:
                flow BY happy1-excit1*0 (a1-a3);
                fhigh BY relax1-conf1*0 (b1-b3);

MODEL PRIORS:   a1-b3~ALF(0,1);
```


Input for EFA

```
                USEVARIABLES = relax1 satis1 conf1 happy1 energ1 excit1;
                ! 6 PA items, 3 low affect, 3 high affect:
                ! relaxed satisfied confident
                ! happy energetic excited

ANALYSIS:       ESTIMATOR = MLR;
                TYPE = EFA 1 3; ! Default Geomin rotation
```


Allowing Cross-Loadings: Log Likelihood and Chi-Square Results for 2 Factors Day 1, N = 215

- EFA log likelihood = -1160.532, MLR chi-square (4) = 1.614 (p-value = 0.8063)
- Bayes with prior N(0,0.01) gives posterior predictive p-value = 0.514
- PSEM with prior ALF(0,1) gives loglikelihood = -1160.547, MLR chi-square(4) = 1.643 (p-value = 0.8011)
  - The "null model" for PSEM is the unrotated EFA model (which has the same LL as rotated): The two log likelihoods should agree (close enough here)
  - PSEM with ALF(0,10) gives exactly the same loglikelihood

Factor Solution using EFA, BSEM, and PSEM Standardized Estimates, Day 1, N = 215

	EFA Geomin		BSEM		PSEM	
RELAX	**0.99**	**-0.19**	**0.92**	-0.11	**0.87**	-0.03
SATIS	**0.95**	0.02	**0.92**	0.06	**0.83**	0.18
CONF	**0.65**	0.13	**0.65**	0.13	**0.57**	**0.24**
HAPPY	**0.47**	**0.53**	**0.31**	**0.67**	**0.39**	**0.62**
ENERG	0.07	**0.85**	-0.05	**0.93**	0.03	**0.88**
EXCIT	-0.01	**0.99**	-0.11	**1.06**	-0.03	**1.00**
Correlation	0.76		0.79		0.69	

- Progression in methodology and acronyms:
  - ESEM (ML, 2009) - BSEM (Bayes, 2012) - PSEM (ML, 2023)

Other Recent Developments

- Asparouhov & Muthén (2021). Expanding the Bayesian structural equation, multilevel and mixture models to logit, negative-binomial and nominal variables. Structural Equation Modeling: A Multidisciplinary Journal, 28:4, 622-637
- Asparouhov & Muthén (2021). Advances in Bayesian model fit evaluation for structural equation models, Structural Equation Modeling: A Multidisciplinary Journal, 28:1, 1-14,
- Asparouhov & Muthén (2021). Bayesian estimation of single and multilevel models with latent variable interactions. Structural Equation Modeling: A Multidisciplinary Journal, 28:2, 314-328
- Asparouhov & Muthén (2020). Comparison of models for the analysis of intensive longitudinal data. Structural Equation Modeling: A Multidisciplinary Journal, 27(2) 275-297